IC Design with Open-Source EDA Tools

Project 1

CMOS Ring Oscillator

Simulation and PVT Analysis using ngspice + Sky130 PDK

Author:	Toney
Process:	SkyWater Sky130 (130 nm CMOS)
Supply Voltage:	1.8 V
Date:	May 2026

Part 1 of 8

Summer IC Design Series

Contents

1 Project Overview

1.1 Objective

This report documents the design, simulation, and analysis of a 3-stage CMOS ring oscillator using open-source EDA tools. The project covers:

1. Writing a SPICE netlist for a CMOS ring oscillator using Sky130 transistor models.
2. Running transient simulation and measuring oscillation frequency with ngspice.
3. Performing PVT (Process, Voltage, Temperature) analysis across worst-case corners.

1.2 Tools and Environment

Tool	**Version**	**Purpose**
ngspice	36	SPICE circuit simulator
Python	3.10.12	Required for volare PDK manager
volare	0.20.6	Sky130 PDK version manager
Sky130 PDK	bdc9412b	130 nm open-source process design kit
nano	6.2	Terminal text editor for netlists

Table 1: Tools used in Project 1

1.3 Project Outcome

Key result: The ring oscillator oscillates at 1.77 GHz at the typical process corner (TT), 27°C, and 1.8V supply. Under worst-case PVT conditions (SS corner, 100°C, 1.62 V), frequency drops to 967 MHz—a 45% reduction, directly demonstrating why Static Timing Analysis (STA) is essential in production silicon.

2 Theory

2.1 What is a Ring Oscillator?

A ring oscillator generates a periodic clock signal with no external input by connecting an odd number of inverters in a loop. Each inverter flips the signal and introduces a propagation delay t_{pd}. Because there is an odd number of inversions, the loop can never settle to a stable state, so it oscillates continuously.

2.1.1 Oscillation Frequency

$$f = \frac{1}{2 \cdot N \cdot t_{pd}}$$

where N is the number of inverter stages (3 here) and t_{pd} is the propagation delay of a single inverter. The factor of 2 accounts for one full HIGH → LOW → HIGH cycle.

2.1.2 Why an Odd Number of Stages?

With an even number of inverters, the loop finds a stable logic state and stops switching. With an odd number, every possible state creates a logical contradiction, forcing the loop to keep toggling forever.

2.2 CMOS Inverter

Each stage is a CMOS inverter: a PMOS transistor pulls the output to VDD when the input is LOW, while an NMOS transistor pulls the output to GND when the input is HIGH. They form a complementary switch—when one is ON the other is OFF.

Transistor	**Type**	**ON when input is**	**Output driven to**
M_p (PMOS)	P-channel	LOW (0V)	VDD = 1.8 V (HIGH)
M_n (NMOS)	N-channel	HIGH (1.8V)	GND = 0V (LOW)

Table 2: CMOS inverter transistor behaviour

2.3 Sky130 PDK

The SkyWater Sky130 PDK is a fully open-source 130 nm CMOS process design kit. Key facts:

- Minimum transistor gate length: $L_{min} = 0.15\mu m$
- Nominal supply voltage: 1.8 V
- Transistors are defined as subcircuits containing internal parasitics. They must be instantiated with the `X` prefix in SPICE, not `M`.

3 Netlist Design

3.1 SPICE Netlist Format

In ngspice, a circuit is described in a plain-text netlist file. Each line declares one component and the nodes it connects to. The general format for a Sky130 MOSFET instance is:

```
Xname drain gate source bulk ModelName W=w L=l
```

3.2 Final Netlist

```
* Ring Oscillator - 3 stage
* Sky130 CMOS Process

* Load the Sky130 PDK tt (typical-typical) corner
.lib /home/toney/pdk/sky130A/libs.tech/ngspice/sky130.lib.spice tt

* Supply voltage
VDD vdd 0 1.8

* Inverter 1 - PMOS + NMOS (X prefix: Sky130 models are subcircuits)
X1_p net1 net3 vdd vdd sky130_fd_pr__pfet_01v8 W=1 L=0.15
X1_n net1 net3 0   0   sky130_fd_pr__nfet_01v8 W=0.5 L=0.15

* Inverter 2
X2_p net2 net1 vdd vdd sky130_fd_pr__pfet_01v8 W=1 L=0.15
X2_n net2 net1 0   0   sky130_fd_pr__nfet_01v8 W=0.5 L=0.15

* Inverter 3
X3_p net3 net2 vdd vdd sky130_fd_pr__pfet_01v8 W=1 L=0.15
X3_n net3 net2 0   0   sky130_fd_pr__nfet_01v8 W=0.5 L=0.15

* Small capacitor to help startup
C1 net1 0 10f
C2 net2 0 10f
C3 net3 0 10f

* Initial condition to kick start oscillation
.ic V(net1)=1.8 V(net2)=0 V(net3)=1.8

* Transient simulation - run for 10ns, step 1ps
.tran 1p 10n

.control
run
plot V(net1) V(net2) V(net3)
.endc
.end
```

Listing 1: ring_oscillator.sp complete netlist

4 Simulation 1: Transient Waveform

4.1 Interpretation

The initial conditions show `net1` and `net3` at 1.8 V and `net2` at 0V—exactly as set by the `.ic` statement. The `vdd#branch` current of $-3.2 \times 10^{-10} A$ (0.32 nA) is the leakage current at room temperature, essentially zero for a healthy digital circuit at 27°C.

A waveform window opens showing three alternating square waves on `net1`, `net2`, and `net3`, each phase-shifted by one inverter delay, confirming the ring oscillator is working correctly.



Figure 1: Transient Waveform of 3-Stage Ring Oscillator (TT Corner)

5 Simulation 2: Frequency Measurement

5.1 Updated Control Block

The control block was updated to add a `meas` command that calculates the period precisely, then derives the frequency:

```
.control
run
plot V(net1) V(net2) V(net3)
* Measure time between 1st and 2nd rising edge crossing 0.9V
meas tran period_val TRIG v(net1) VAL=0.9 RISE=1 TARG v(net1) VAL=0.9
    RISE=2
let freq_val = 1 / period_val
print freq_val
.endc
```

5.2 Results



Figure 2: Waveform with Period Measurement (TT Corner, 1.8V, 27°C)

Variable	Raw value	Human-readable	Meaning
period_val	5.640×10^{-10}	564 ps	Time for one full cycle
freq_val	1.773×10^{9} Hz	1.77 GHz	Oscillations per second

Table 3: Decoded frequency measurement

At the typical process corner (TT), room temperature (27°C), and nominal 1.8 V supply, the 3-stage ring oscillator runs at 1.77 GHz.

6 PVT Analysis

6.1 What is PVT Analysis?

Real silicon faces manufacturing variations, voltage fluctuations, and temperature extremes. PVT analysis systematically tests Process (P), Voltage (V), and Temperature (T).

6.2 Process Corners

The foundry tests extreme cases, called corners:

- **tt (Typical-Typical):** Normal NMOS, Normal PMOS
- **ff (Fast-Fast):** Fast NMOS, Fast PMOS (max speed, high leakage)
- **ss (Slow-Slow):** Slow NMOS, Slow PMOS (min speed, low leakage)

6.3 Worst-Case Corner: SS, 100°C, 1.62 V

This simulates the absolute worst case: slow transistors, maximum operating temperature, and a 10% voltage droop. Changes to the netlist:

```
.lib /home/toney/pdk/sky130A/libs.tech/ngspice/sky130.lib.spice ss
.option temp=100
VDD vdd 0 1.62
```

6.4 SS Corner Results

Corner	Voltage	Temp	Frequency	vs. TT
TT (nominal)	1.8 V	27°C	1.77 GHz	Baseline
SS (worst case)	1.62 V	100°C	967 MHz	-45%

Table 4: PVT analysis results summary

6.5 Analysis: Why the Frequency Dropped

- **Lower voltage (1.62 V):** Reduces gate overdrive ($V_{GS} - V_{TH}$), reducing drain current I_{on}. Charging takes longer ($t = \frac{C \cdot \Delta V}{I}$).
- **High temperature (100°C):** Thermal lattice scattering increases, reducing electron and hole mobility. Leakage jumps exponentially.
- **SS process corner:** Transistors have inherently higher resistance and slower switching speeds based on fabrication variance.



Figure 3: Waveform at Worst-Case PVT (SS Corner, 1.62V, 100°C)

7 Key Learnings

1. **Ring oscillator principle:** An odd number of inverters in a feedback loop prevents any stable state, producing perpetual oscillation.
2. **Sky130 subcircuit models:** Modern PDKs wrap transistors in subcircuits to include parasitics, requiring the `X` prefix in SPICE.
3. **Temperature and leakage:** High temperature causes an exponential increase in subthreshold leakage (jumping from 0.32 nA to 41 μA).
4. **STA motivation:** The 45% frequency drop from TT to worst-case is the quantitative reason Static Timing Analysis exists.

8 Conclusion

This project successfully designed, simulated, and analysed a 3-stage CMOS ring oscillator using open-source tools. The 45% frequency reduction across PVT corners demonstrates the importance of PVT-aware design. A chip designed to run at 1.5 GHz assuming typical conditions would fail in the field under realistic thermal and voltage stress.

Task	**Status**
ngspice 36 installed	✓Complete
Sky130 PDK installed	✓Complete
Ring oscillator netlist written	✓Complete
Transient simulation (waveform)	✓Complete
Frequency measurement (1.77GHz)	✓Complete
PVT analysis (SS worst case)	✓Complete

Table 5: Project completion checklist